Traditional Logic

CS 3234: Logic and Formal Systems

Martin Henz and Aquinas Hobor

August 19, 2010

Generated on Monday 23 August, 2010, 14:10

1 Motivation

Our first deductive system expresses relationships between classes of things. Each class is represented by a noun naming the class, or by an adjective that describes a property that distinguishes the members of the class from non-members. These nouns or adjectives denoting classes of things are called *categorical terms*, or simply *terms*.

For example, the term `animals` refers to the class of things that contains all animals.

As another example, the term `brave` refers to the class of persons that we consider brave. Note that we are not very concerned with the difficulty to deliberate the attribute of bravery in individual cases. In logic, we do not spend much time arguing whether a particular person, say Socrates, is brave or not. Instead, we assume that it is always possible to univocally attest the given attribute to a given person or not.

Categorical terms constitute the basic unit of meaning in our first deductive system, which is therefore called term logic (sometimes also categorical logic). In the history of philosophy, term logic plays a prominent role, because many arguments that appear in everyday discourse (and also in political statements, philosophical treatises, etc) can be analysed and verified using term logic. In fact, term logic has been investigated extensively by the Greek philosopher Aristotle (384–322 BCE) already, whose treatise *Prior Analytics* is considered the earliest study in formal logic and was widely accepted as the definitive approach to deductive reasoning until the 19^{th}century.

To understand the concerns of term logic, consider the following argument.

Example 1.

All humans are mortal.
All Greeks are humans.
Therefore, all Greeks are mortal.

This argument appears acceptable and consistent with our intuition. But what exactly makes it acceptable, and what makes the following "argument" unacceptable?

Example 2.

> *All cats are predators.*
> *Some animals are cats.*
> *Therefore, all animals are predators.*

Of course, we know that the conclusion of the latter "argument" is false, so we know that there must be something wrong with the argument. However, even if we do not know anything about the subject matter, arguments of the first kind are acceptable. Consider for example:

Example 3.

> *All slack track systems are caterpillar systems.*
> *All Christie suspension systems are slack track systems.*
> *Therefore, all Christie suspension systems are caterpillar systems.*

Even if you have no knowledge of caterpillar suspension systems, the argument appears sound. It appeals to you because of the way the categorical terms are arranged in the argument, rather than the choice of the categorical terms, themselves. Our study of logic focuses on what arguments are acceptable, or *hold*, due to their form alone. In order to do so, it is important to precisely define what an argument consists of, what we mean by *validity*, and what methods we can use to demonstrate validity.

2 Terms and their Semantics

In our reasoning framework, we define categorical terms as members of a particular *data type*, `Term`, whose elements name classes of entities. For example, `animals` may be an element of `Term`, representing all entities which are considered animals.

> In parallel with the discussion of the topic, we will introduce a formal system that illustrates the particular style of reasoning, supported by a software system called Coq. Throughout this book, we make use of Coq as a *proof assistant*, helping us formalize arguments and assisting us in the construction of proofs, as we advance through a sequence of more and more complex and powerful logics. In Coq, we can write scripts that define the logic

```
Module TraditionalLogic.
```

> In Coq, we can define the type `Term` as follows.

```
Parameter Term : Type.
```

Now we can populate the type `Term` with particular instances.

```
Parameter humans : Term.
Parameter Greeks : Term.
Parameter mortal : Term.
```

These declarations simply express that the entities `humans`, `Greeks`, and `mortal` are all terms: `humans` $\in$ `Term`, `Greeks` $\in$ `Term`, and `mortal` $\in$ `Term`.

In this section, we clarify the relationship between terms what they represent. A particular meaning $\mathcal{M}$, also called *model*, fixes a universe of discourse, denoted $U^{\mathcal{M}}$, and for every element $t \in \texttt{Term}$, a set $t^{\mathcal{M}}$, where $t^{\mathcal{M}} \subseteq U^{\mathcal{M}}$.[1]

For example, for reasoning about living beings such as cats, humans, Greeks, and so on, we may choose a meaning $\mathcal{M}$ whose universe $U^{\mathcal{M}}$ is the set of all living beings, whose $\texttt{cat}^{\mathcal{M}}$ the set of all cats, whose $\texttt{humans}^{\mathcal{M}}$ the set of all humans, and so on. However, for the same terms, we may consider a different meaning $\mathcal{M}'$ whose universe $U^{\mathcal{M}'}$ is a set of labeled playing cards, whose $\texttt{cat}^{\mathcal{M}'}$ contains those cards that display cats, whose $\texttt{humans}^{\mathcal{M}'}$ contains those cards that display humans, and so on.

Example 4. *Consider the following set of terms:*

$$\texttt{Term} = \{\texttt{even}, \texttt{odd}, \texttt{belowfour}\}$$

We could choose a meaning $\mathcal{M}_1$, where $U^{\mathcal{M}_1} = \mathbb{N}$, and the "obvious" meaning $\texttt{even}^{\mathcal{M}_1} = \{0, 2, 4, \ldots\}$, $\texttt{odd}^{\mathcal{M}_1} = \{1, 3, 5, \ldots\}$, and $\texttt{belowfour}^{\mathcal{M}_1} = \{0, 1, 2, 3\}$.

Alternatively, we could choose a meaning $\mathcal{M}_2$ where $U^{\mathcal{M}_2} = \{0, 1, 2, 3, 4, 5, 6\}$, and the "obvious" meaning $\texttt{even}^{\mathcal{M}_2} = \{0, 2, 4, 6\}$, $\texttt{odd}^{\mathcal{M}_2} = \{1, 3, 5\}$, and $\texttt{belowfour}^{\mathcal{M}_2} = \{0, 1, 2, 3\}$.

However, no one can prevent us from choosing an unexpected meaning $\mathcal{M}_3$, in which $U^{\mathcal{M}_3} = \{a, b, c, \ldots, z\}$, $\texttt{even}^{\mathcal{M}_3} = \{a, e, i, o, u\}$, $\texttt{odd}^{\mathcal{M}_3} = \{b, c, d, \ldots\}$, and $\texttt{belowfour}^{\mathcal{M}_3} = \emptyset$.

3 Categorical Propositions

Our term logic allows us to express relationships between two categorical terms. For example, we may want to investigate the statement

[1] More formally, we can define a semantics $\mathcal{M}$ of a set of terms `Term` as a pair $(U, \textit{interprete})$, where U is a set, and *interprete* is a function

$$\textit{interprete} : \texttt{Term} \to \mathcal{P}(U)$$

where $\mathcal{P}(U)$ denotes the set of all subsets of U. Thus, for a given `Term` t, $\textit{interprete}(t)$, denoted by $t^{\mathcal{M}}$, is a subset of U.

`All cats are predators`

This statement expresses a relationship between the terms `cats` and `predators`, saying that every thing that is included in the class represented by `cats` is also included in the class represented by `predators`.

Such statements are called categorical propositions. The first categorical term in the proposition (in our case `cats`) is called the subject of the proposition, and the second term (in our case `predators`) is called its object. Categorical propositions of the form

All t_1 are t_2

where t_1 and t_2 are terms, are called universal affirmative propositions.

Similarly, we provide for universal negative propositions such as `No Greeks are cats`, particular affirmative propostions such as `Some animals are cats`, and particular negative propositions such as `Some cats are not brave`.

Thus, categorical propositions come in four forms, depending on the *quantity* (universal or particular), and *quality* (affirmative or negative).

We first provide for the two quantities as constants of the type `Quantity`.

```
Record Quantity : Type :=
  universal : Quantity
| particular : Quantity.
```

Similarly, the two qualities are constants of type `Quality`.

```
Record Quality : Type :=
  affirmative : Quality
| negative : Quality.
```

Now, data structures of type `CategoricalProposition` are constructed from a `Quantity`, a `Quality`, a subject `Term` and an object `Term`.

```
Record CategoricalProposition : Type :=
cp {
  quantity : Quantity;
  quality : Quality;
  subject : Term;
  object : Term
}.
```

It is convenient to be able to give a name to particular propositions, using the keyword `Definition`.

```
Definition HumansMortal : CategoricalProposition :=
  cp universal affirmative humans mortal.
```

Note that the *constructor* cp is used to form the proposition. As a matter of syntactic convenience, we would like to write categorical propositions in the same way as we write them in English, namely:

```
All humans are mortal
```

instead of

```
cp universal affirmative humans mortal
```

This is accomplished using the following `Notation` declarations.

```
Notation "'All' subject 'are' object " :=
   (cp universal affirmative subject object) (at level 50).

Notation "'No' subject 'are' object " :=
   (cp universal negative subject object) (at level 50).

Notation "'Some' subject 'are' object " :=
   (cp particular affirmative subject object) (at level 50).

Notation "'Some' subject 'are' 'not' object " :=
   (cp particular negative subject object) (at level 50).
```

Now we can write:

```
Definition HumansMortal2 : CategoricalProposition :=
   All humans are mortal.

Parameter cats : Term.

Definition NoExample: CategoricalProposition :=
   No Greeks are cats.

Parameter animals : Term.

Definition SomeExample: CategoricalProposition :=
   Some animals are cats.

Parameter brave : Term.

Definition SomeNotExample: CategoricalProposition :=
   Some cats are not brave.
```

Note that all propositions deal with terms that describe the relationship between classes of entities, not individual entities. In order to express that a particular entity, say Socrates, is included in a class, say Greek, one would need to form a term such as "people with the name Socrates", and then state the universal affirmative proposition "All people with the name Socrates are Greek".[2]

4 Semantics of Categorical Propositions

Recall that terms represent subsets of a particular domain of discourse. Categorical propositions describe relationships between these sets. For example, the proposition

`All humans are mortal`

says that the set `humans` is a subset of the set `mortal`. Once we fix the sets that the terms represent, it is clear which propositions hold and which don't.

Thus, we can define the meaning of the categorical proposition with respect to a model $\mathcal{M}$ as follows:

$$(\texttt{All}\ \textit{subject}\ \texttt{are}\ \textit{object})^{\mathcal{M}} = \begin{cases} T & \text{if } \textit{subject}^{\mathcal{M}} \subseteq \textit{object}^{\mathcal{M}}, \\ F & \text{otherwise} \end{cases}$$

Here T and F represent the logical truth values *true* and *false*, respectively. We visualize a universal affirmative proposition such as `All Greeks are mortal` using a *Venn diagram* as follows:



The darkest shading indicates an area that does not contain any entities, if the proposition is true.

[2] Or should it be a particular affirmative proposition "Some people with the name Socrates are Greek"? This question gives you a hint of the philosophical difficulties posed by pushing traditional logic beyond classes of entities.

Similarly, we can define the meaning of the other categorical propositions:

$$(\texttt{No}\ \textit{subject}\ \texttt{are}\ \textit{object})^{\mathcal{M}} = \begin{cases} T & \text{if } \textit{subject}^{\mathcal{M}} \cap \textit{object}^{\mathcal{M}} = \emptyset, \\ F & \text{otherwise} \end{cases}$$

Again, the darkest shading in the diagram for `No Greeks are cats` indicates an empty area.



The meaning of particular affirmative propositions is given by

$$(\texttt{Some}\ \textit{subject}\ \texttt{are}\ \textit{object})^{\mathcal{M}} = \begin{cases} T & \text{if } \textit{subject}^{\mathcal{M}} \cap \textit{object}^{\mathcal{M}} \neq \emptyset, \\ F & \text{otherwise} \end{cases}$$

and visualized through the following diagram.



The darkest region in the diagram for `Some humans are Greeks` now represents an area that contains at least one entity.

Finally, the meaning of particular negative propositions is given by

$$(\texttt{Some}\ \textit{subject}\ \texttt{are not}\ \textit{object})^{\mathcal{M}} = \begin{cases} T & \text{if } \textit{subject}^{\mathcal{M}} / \textit{object}^{\mathcal{M}} \neq \emptyset, \\ F & \text{otherwise} \end{cases}$$

A proposition such as `Some Greeks are not vegetarians` is visualized by



where the darkest region represents an area that contains at least one entity.

5 Axioms, Lemmas and Proofs

We would like to be able to state that a particular categorical proposition holds, so that we can later make use of it as a fact. In logic, propositions that are assumed to hold are called *axioms*.

In order to refer to an axiom later on, we allow ourselves to give it a name. For example, we can state the mortality of humans and the humanity of Greeks as follows.

Axiom 1 (HumansMortality). *The proposition `All humans are mortal` holds.*

Axiom 2 (GreeksHumanity). *The proposition `All Greeks are humans` holds.*

In Coq, asserting a proposition is done using *axioms* of the form

`Axiom` *name* `:` *proposition*.

As *proposition*, we need to use a particular built-in type of Coq called `Prop`; we cannot assert a `CategoricalProposition` as a fact. Therefore, we define a function `holds` that turns a given `CategoricalProposition` into a `Prop`, thereby asserting that it holds.

```
Parameter holds : CategoricalProposition -> Prop.
```

Now we can assert the mortality of humans and the humanity of Greeks as axioms.

```
Axiom HumansMortality: holds (All humans are mortal).

Axiom GreeksHumanity: holds (All Greeks are humans).
```

We introduce a graphical notation for axioms, where a horizontal bar is used to separate possible premises above the bar from the conclusion below the bar. Since a fact holds regardless of any premise, we display it as follows:

$$\frac{}{\texttt{All humans are mortal}}\,[\text{HumansMortality}]$$

Lemmas are affirmations that follow from all known facts. A lemma must be followed by a proof that demonstrates how it follows from known facts. We show this mechanism by stating the mortality of humans as a lemma, and prove it by simply applying the axiom *HumansMortality*.

Lemma 1. *The proposition `All humans are mortal` holds.*

We prove this lemma by simply invoking the axiom *HumansMortality*.

Proof.

$$\frac{}{\texttt{All humans are mortal}}\,[\text{HumansMortality}]$$

□

```
Lemma HumansMortality2: holds (All humans are mortal).

Proof.
apply HumansMortality.
Qed.
```

6 What do Axioms, Lemmas and Proofs Mean?

According to Section 2, we are free to fix a model $\mathcal{M}$, which selects a subset of our universe for each term. However, such a semantics may or may not meet the requirements posed by a given axiom. For example, if we choose $U^{\mathcal{M}} = \{0, 1\}$, $\texttt{humans}^{\mathcal{M}} = \{0\}$, and $\texttt{mortal}^{\mathcal{M}} = \{1\}$, then clearly the proposition

$$\texttt{All humans are mortal}$$

does not hold.

By asserting an axiom A, we are focusing our attention to only those models $\mathcal{M}$ for which $A^{\mathcal{M}} = T$. The lemmas that we prove while utilizing an axiom only hold in the models in which the axiom holds.

The ability of using an axiom in proofs comes at a price; the proof is valid only for those models in which the axiom holds. Thus, in a sense, our reasoning becomes weaker when we assert an axiom.

A proposition is called *valid*, if it holds in all models.

Exercise 1. *Is the proposition*

```
All humans are humans
```

valid? Use the semantics of categorical propositions (Section 4) in your argument.

Exercise 2. *Is the proposition*

```
Some humans are humans
```

valid? Use the semantics of categorical propositions (Section 4) in your argument.

7 Complement

The terms that we encountered so far consist of single words such as `Greeks`. We call such terms *primitive*. Since terms describe subsets S of a given universe U, it is natural to provide an operation on terms that selects the complement of S, denoted as U/S. For this, we introduce an operation `non` that is applied to a `Term` and results in a `Term`. Thus, if our universe consists of all living things, then the `Term`

```
non Greeks
```

denotes those living things that are not Greeks. Note that the operator `non` is applied to a `Term` and returns a `Term`. Terms that are constructed using such an operator are called *compound* terms.

Of course, for any compound term, we can define a primitive term that is equal to it. For example, the following definition introduces a primitive term `immortal`.

Definition 1 (ImmortalDef)**.** *The term `immortal` is considered equal to the term `non mortal`.*

```
Parameter non: Term -> Term.
```

We can apply the `non` operation to a term, for example to `mortal`:

```
Definition immortal: Term := non mortal.

Definition SomeHumansAreImmortal : CategoricalProposition :=
   Some humans are immortal.
```

Since the complement of the complement of any set is the set itself, we are compelled to assert the following axiom.

Axiom 3 (NonNon)**.** *For any term t, the term `non non` t is considered equal to t.*

```
Axiom NonNon: forall t, non (non t) = t.
```

In graphical notation, the axiom can be depicted as follows.

$$\frac{\cdots t\ \cdots}{\cdots \texttt{non non}\ t \cdots}\,[\text{NNI}]$$

$$\frac{\cdots \texttt{non non}\ t\ \cdots}{\cdots t \cdots}\,[\text{NNE}]$$

where the $\cdots$ allow for any context. Here is an example instance of the first rule.

$$\frac{\texttt{Some}\ t_1\ \texttt{are}\ t_2}{\texttt{Some non non}\ t_1\ \texttt{are}\ t_2}$$

In these rules, the proposition above the bar states premises, and the formula below the bar states the conclusion. We can use the last rule to prove `Some non non` t_1' `are` t_2'; then we are left with the obligation to prove `Some` t_1' `are` t_2'.

The following lemma is proven with the use of this axiom, in addition to the previous axiom `HumansMortality`.

Lemma 2. *The proposition `All humans are non immortal` holds.*

We use our graphical notation to write down the proof for this lemma:

Proof.

$$\dfrac{\dfrac{\dfrac{}{\texttt{All humans are mortal}}\,[\text{HumansMortality}]}{\texttt{All humans are non non mortal}}\,[\text{NNI}]}{\texttt{All humans are non immortal}}\,[\text{ImmortalDef}]$$

□

When the proofs get larger, they often do not fit on a single page, and become difficult to draw. Therefore, we introduce an alternative text-based notation for proofs, where each proposition is given in a numbered line, and the rule that justifies the proposition is given on the right. For example, the text-based notation for the proof above reads as follows.

Proof.

1	`All humans are mortal`	HumansMortality
2	`All humans are non non mortal`	NNI 1
3	`All humans are non immortal`	ImmortalDef 2

□

The third column justifies the proposition, citing the axiom or definition used, if necessary referring to a previous line using its number.

```
Lemma HumansNonImmortality: holds (All humans are (non immortal)).

Proof.
unfold immortal.
rewrite NonNon.
apply HumansMortality.
Qed.
```

Note that the directive `unfold` unfolds the definition of `immortal`, as given above. Now we find a double application of `non`, which presents us with an opportunity to rewrite the proposition using the equality asserted by the axiom `NonNon`. Finally, an application of `HumansMortality` completes the proof.

8 Immediate Inferences

This section introduces transformation operations on propositions, whose validity we intend to study.

8.1 Conversion

Consider the operation `convert`, that switches subject and object of a proposition. Thus we can define:

Definition 2 (ConvDef). *For all terms t_1 and t_2, we define*

$$\begin{aligned} \mathit{convert}(\mathit{All}\ t_1\ \mathit{are}\ t_2) &= \mathit{All}\ t_2\ \mathit{are}\ t_1 \\ \mathit{convert}(\mathit{Some}\ t_1\ \mathit{are}\ t_2) &= \mathit{Some}\ t_2\ \mathit{are}\ t_1 \\ \mathit{convert}(\mathit{No}\ t_1\ \mathit{are}\ t_2) &= \mathit{No}\ t_2\ \mathit{are}\ t_1 \\ \mathit{convert}(\mathit{Some}\ t_1\ \mathit{are\ not}\ t_2) &= \mathit{Some}\ t_2\ \mathit{are\ not}\ t_1 \end{aligned}$$

In Coq, `convert` is defined as a function as follows.

```
Definition convert: CategoricalProposition -> CategoricalProposition :=
  fun x : CategoricalProposition => match x with
    | cp quantity quality subject object =>
      cp quantity quality object subject
  end.
```

Note that not all conversions preserve the meaning of propositions; if `All Greeks are humans` holds in a model, then `All humans are Greeks` may or may not hold in the same model.

```
Definition AllGreeksHumansConverted  :=
      convert (All Greeks are humans).
```

Conversion does preserve the meaning of particular affirmative and universal negative propositions, as expressed by the following axioms.

Axiom 4 (ConvE1). *If, for some terms t_1 and t_2, the proposition*

$$\mathit{convert}(\mathit{Some}\ t_1\ \mathit{are}\ t_2)$$

holds, then the proposition

$$\mathit{Some}\ t_1\ \mathit{are}\ t_2$$

also holds.

Axiom 5 (ConvE2)**.** *If, for some terms t_1 and t_2, the proposition*

$$\texttt{convert}(\texttt{No}\ t_1\ \texttt{are}\ t_2)$$

holds, then the proposition

$$\texttt{No}\ t_1\ \texttt{are}\ t_2$$

also holds.

In graphical notation, two rules correspond to the two cases.

$$\frac{convert(\texttt{Some}\ t_1\ \texttt{are}\ t_2)}{\texttt{Some}\ t_1\ \texttt{are}\ t_2}\,[\mathrm{ConvE}_1]$$

$$\frac{convert(\texttt{No}\ t_1\ \texttt{are}\ t_2)}{\texttt{No}\ t_1\ \texttt{are}\ t_2}\,[\mathrm{ConvE}_2]$$

Exercise 3. *Use Venn diagrams and the definitions in Section 2 to argue why Axioms 4 and 5 are valid.*

In Coq, the two axioms are formulated as follows.

```
Axiom ConvE1:
   forall subject object,
   holds (convert (Some subject are object))
   ->
   holds (Some subject are object).

Axiom ConvE2:
   forall subject object,
   holds (convert (No subject are object))
   ->
   holds (No subject are object).
```

As an example of conversion in action, consider the following axiom.

Axiom 6 (SomeAnimalsAreCats)**.** *The proposition `Some animals are cats` holds.*

```
Axiom SomeAnimalsAreCats : holds (Some animals are cats).
```

We can use conversion, to prove `Some cats are animals` as follows.

Lemma 3. *The proposition* `Some cats are animals` *holds.*

Proof.

$$\cfrac{\cfrac{\cfrac{}{\texttt{Some animals are cats}}\,[\text{SomeAnimalsAreCats}]}{\texttt{convert(Some cats are animals)}}\,[\text{ConvDef}]}{\texttt{Some cats are animals}}\,[\text{ConvE}_1]$$

□

The same proof follows in text-based notation.

Proof.

1	`Some animals are cats`	SomeAnimalsAreCats
2	`convert(Some cats are animals)`	ConvDef 1
3	`Some cats are animals`	ConvE_1 2

□

Here is the Coq version of this proof.

```
Lemma SomeCatsAreAnimals : holds (Some cats are animals).

Proof.
apply ConvE1.
unfold convert.
apply SomeAnimalsAreCats.
Qed.
```

Similarly, the validity of conversion of universal negative propositions is used in the proof below.

Axiom 7 (NoGreeksAreCats)**.** *The proposition* `No Greeks are cats` *holds.*

We can use conversion, to prove `No cats are Greeks` as follows.

Lemma 4 (NoCatsAreGreeks)**.** *The proposition* `No cats are Greeks` *holds.*

Proof.

$$\dfrac{\dfrac{\dfrac{}{\texttt{No Greeks are cats}}\,[\text{NoGreeksAreCats}]}{\texttt{convert(No cats are Greeks)}}\,[\text{ConvDef}]}{\texttt{No cats are Greeks}}\,[\text{ConvE}_2]$$

□

In text-based notation, the proof goes as follows:

Proof.

1	`No Greeks are cats`	NoGreeksAreCats
2	`convert(No cats are Greeks)`	ConvDef 1
3	`No cats are Greeks`	ConvE_2 2

□

And, for completeness, the same in Coq:

```
Axiom NoGreeksAreCats : holds (No Greeks are cats).

Lemma NoCatsAreGreeks : holds (No cats are Greeks).

Proof.
apply ConvE2.
unfold convert.
apply NoGreeksAreCats.
Qed.
```

The previous two proofs follow a similar pattern: prove the instance using `ConvE1` or `ConvE2` and unfold conversion. In Coq, such a pattern can be expressed as a *tactic*. We have pre-defined tactics `eliminateConversion1` and `eliminateConversion2` so that we can simplify the proof of `No cats are Greeks` as follows.

```
Lemma NoCatsAreGreeks2 : holds (No cats are Greeks).

Proof.
eliminateConversion2.
apply NoGreeksAreCats.
Qed.
```

8.2 Contraposition

Similar to conversion, we define an operation called *contraposition*, which reverses the roles of subject and object and forms the complement of both. Thus, the contraposition of `All Greeks are humans` is `All non humans are non Greeks`.

Definition 3 (ContrDef). *For all terms t_1 and t_2, we define*

$$
\begin{aligned}
\mathit{contrapose}(\mathit{All}\ t_1\ \mathit{are}\ t_2) &= \mathit{All}\ \mathit{non}\ t_2\ \mathit{are}\ \mathit{non}\ t_1\\
\mathit{contrapose}(\mathit{Some}\ t_1\ \mathit{are}\ t_2) &= \mathit{Some}\ \mathit{non}\ t_2\ \mathit{are}\ \mathit{non}\ t_1\\
\mathit{contrapose}(\mathit{No}\ t_1\ \mathit{are}\ t_2) &= \mathit{No}\ \mathit{non}\ t_2\ \mathit{are}\ \mathit{non}\ t_1\\
\mathit{contrapose}(\mathit{Some}\ t_1\ \mathit{are}\ \mathit{not}\ t_2) &= \mathit{Some}\ \mathit{non}\ t_2\ \mathit{are}\ \mathit{not}\ \mathit{non}\ t_1
\end{aligned}
$$

```
Definition contrapose: CategoricalProposition ->
                       CategoricalProposition :=
  fun x : CategoricalProposition => match x with
    | cp quantity quality subject object =>
      cp quantity quality (non object) (non subject)
  end.
```

```
Definition AllGreeksHumansContraposed  :=
      contrapose (All Greeks are humans).
```

Contraposition is a valid operation on universal affirmative and particular negative propositions.

Axiom 8 (ContrE1). *If, for some terms t_1 and t_2, the proposition*

$$\mathit{contrapose}(\mathit{All}\ t_1\ \mathit{are}\ t_2)$$

holds, then the proposition

$$\mathit{All}\ t_1\ \mathit{are}\ t_2$$

also holds.

Axiom 9 (ContrE2). *If, for some terms t_1 and t_2, the proposition*

$$\mathit{contrapose}(\mathit{Some}\ t_1\ \mathit{are}\ \mathit{not}\ t_2)$$

holds, then the proposition

$$\mathit{Some}\ t_1\ \mathit{are}\ \mathit{not}\ t_2$$

also holds.

```
Axiom ContrE1:
   forall subject object,
   holds (contrapose (All subject are object))
   ->
   holds (All subject are object).

Axiom ContrE2:
   forall subject object,
   holds (contrapose (Some subject are not object))
   ->
   holds (Some subject are not object).
```

In graphical notation, two rules correspond to the two cases.

$$\frac{\mathit{contrapose}(\texttt{All } t_1 \texttt{ are } t_2)}{\texttt{All } t_1 \texttt{ are } t_2}\,[\mathrm{ContrE}_1]$$

$$\frac{\mathit{contrapose}(\texttt{Some } t_1 \texttt{ are not } t_2)}{\texttt{Some } t_1 \texttt{ are not } t_2}\,[\mathrm{ContrE}_2]$$

Exercise 4. *Use contraposition to prove*

`All (non animals) are (non cats)`

using `All cats are animals` *as axiom. Give your proof in graphical and text-based notation.*

Sometimes you may not want to prove a lemma immediately. In Coq, you can write `admit` to "pretend" the current goal to be proven. Of course, proofs that contain `admit` are not valid. We use `admit` in order to pose exercises.

Coq Exercise 1. *Replace* `admit` *in the following to obtain a valid proof.*

```
Axiom AllCatsAreAnimals : holds (All cats are animals).

Lemma AllNonAnimalsAreNonCats : holds (All (non animals) are (non cats)).

Proof.
admit.
Qed.
```

Hint: You may need to use `Axiom NonNon` *from Section 7.*

Exercise 5. *As an example for particular negative propositions, use contraposition to prove* `Some non cats are not non animals` *with the help of an axiom* `Some animals are not cats`*. Give your proof in graphical and text-based notation.*

Coq Exercise 2. *Replace* `admit` *in the following to obtain a valid proof.*

```
Axiom SomeAnimalsAreNotCats : holds (Some animals are not cats).

Lemma SomeNonCatsAreNotNonAnimals : holds (Some (non cats) are not (non animals)).

Proof.
admit.
Qed.
```

As with conversion, we have defined corresponding tactics `eliminateContraposition1` and `eliminateContraposition2` which allow us to rewrite the proof.

Coq Exercise 3. *Use these tactics to simplify the proof of* `Some (non cats) are not (non animals)`*.*

```
Lemma SomeNonCatsAreNotNonAnimals2 : holds (Some (non cats) are not (non animals)).

Proof.
admit.
Qed.
```

8.3 Obversion

The final operation on propositions is called *obversion*. This operation complements the object and turns affirmative propositions into negative propositions, and vice versa. Thus, applying obversion to the universal affirmative proposition `All Greeks are humans` results in the universal negative proposition `No Greeks are non humans`, and the particular negative proposition `Some animals are not cats` results in the particular affirmative proposition `Some animals are non cats`.

Definition 4 (ObvDef)**.** *For all terms t_1 and t_2, we define*

$$\begin{aligned}
\mathtt{obvert}(\mathtt{All}\ t_1\ \mathtt{are}\ t_2) &= \mathtt{No}\ t_1\ \mathtt{are}\ \mathtt{non}\ t_2 \\
\mathtt{obvert}(\mathtt{Some}\ t_1\ \mathtt{are}\ t_2) &= \mathtt{Some}\ t_1\ \mathtt{are}\ \mathtt{not}\ \mathtt{non}\ t_2 \\
\mathtt{obvert}(\mathtt{No}\ t_1\ \mathtt{are}\ t_2) &= \mathtt{All}\ t_1\ \mathtt{are}\ \mathtt{non}\ t_2 \\
\mathtt{obvert}(\mathtt{Some}\ t_1\ \mathtt{are}\ \mathtt{not}\ t_2) &= \mathtt{Some}\ t_1\ \mathtt{are}\ \mathtt{non}\ t_2
\end{aligned}$$

We first introduce a means to obtain the opposite of a quality.

```
Definition complement: Quality -> Quality :=
  fun x : Quality => match x with
    | affirmative => negative
    | negative => affirmative
  end.
```

Now we can define a function `obvert`.

```
Definition obvert: CategoricalProposition -> CategoricalProposition :=
  fun x : CategoricalProposition => match x with
    | cp quantity quality subject object
      => cp quantity (complement quality) subject (non object)
  end.
```

Obversion is valid for all kinds of propositions.

Axiom 10 (ObvE)**.** *If, for some proposition p*

$$\mathtt{obvert}(p)$$

holds, then the proposition p also holds.

```
Axiom ObvE :
   forall catprop, holds (obvert catprop) -> holds catprop.
```

In our rule notation, we write the axiom as depicted below:

$$\frac{\mathit{obvert}(p)}{p}\,[\text{ObvE}]$$

Example 5. *We prove* `Some humans are not non vegetarians` *from* `Some humans are vegetarians`*.*

Axiom 11 (SomeHumansVegetarians)**.** *The proposition* `Some humans are vegetarians` *holds.*

Lemma 5 (NNVeg)**.** *The proposition* `Some humans are not non vegetarians` *holds.*

Proof.

$$
\dfrac{\dfrac{\dfrac{\dfrac{}{\texttt{Some humans are vegetarians}}\,[\text{SomeHumansVegetarians}]}{\texttt{Some humans are non non vegetarians}}\,[\text{NNI}]}{\texttt{obvert(Some humans are not non vegetarians)}}\,[\text{ObvDef}]}{\texttt{Some humans are not non vegetarians}}\,[\text{ObvE}]
$$

□

We see here already that the graphical notation becomes inconvenient when proofs reach a certain size. The text-based alternative is more suitable for larger proofs.

Proof.

1	`Some humans are vegetarians`	SomeHumansVegetarians
2	`Some humans are non non vegetarians`	NNI 1
3	`obvert(Some humans are not non vegetarians)`	ObvDef 2
4	`Some humans are not non vegetarians`	ObvE 3

□

In Coq, we use `rewrite NonNon`, as shown before.

```
Parameter vegetarians : Term.

Axiom SomeHumansVegetarians : holds (Some humans are vegetarians).

Lemma SomeHumansAreNotNonVegetarians :
   holds (Some humans are not (non vegetarians)).

Proof.
apply ObvE.
unfold obvert.
unfold complement.
rewrite NonNon.
apply SomeHumansVegetarians.
Qed.
```

Again, a pre-defined tactic `eliminateObversion` simplifies the proofs.

```
Lemma SomeHumansAreNotNonVegetarians2 :
   holds (Some humans are not (non vegetarians)).

Proof.
eliminateObversion.
rewrite NonNon.
apply SomeHumansVegetarians.
Qed.
```

8.4 Combinations

We can combine our three operations—conversion, contraposition and obversion—to obtain further results.

Lemma 6 (SomeNon)**.** *For all terms t_1 and t_2, if the proposition `Some non` t_1 `are non` t_2 holds, then the proposition `Some non` t_2 `are not` t_1 also holds.*

A lemma of the form "If p_1 then p_2" is valid, if in every model in which the proposition p_1 holds, the proposition p_2 also holds.

Exercise 6. *Argue that Lemma 6 is valid, using the definitions of the meaning of the respective propositions (Section 4). Use a Venn diagram to illustrate your argument.*

In order to prove a lemma of the form `If` p_1, `then` p_2, we need to obtain p_2, using p_1 as a *premise.*

Proof.

$$\dfrac{\dfrac{\dfrac{\dfrac{\dfrac{}{\texttt{Some non } t_1 \texttt{ are non } t_2}\,[\text{premise}]}{\texttt{convert(Some non } t_2 \texttt{ are non } t_1\texttt{)}}\,[\text{ConvDef}]}{\texttt{Some non } t_2 \texttt{ are non } t_1}\,[\text{ConvE}_1]}{\texttt{obvert(Some non } t_2 \texttt{ are not } t_1\texttt{)}}\,[\text{ObvDef}]}{\texttt{Some non } t_2 \texttt{ are not } t_1}\,[\text{ObvE}]$$

□

Proof.

1	`Some non` t_1 `are non` t_2	premise
2	`convert(Some non` t_2 `are non` t_1`)`	ConvDef 1
3	`Some non` t_2 `are non` t_1	ConvE_1 2
4	`obvert(Some non` t_2 `are not` t_1`)`	ObvDef 3
5	`Some non` t_2 `are not` t_1	ObvE 4

□

Note that a premise plays the same role as an axiom in a prove; it can be stated without a need for further proof.

```
Lemma SomeNon: forall subject object,
     holds (Some non object are non subject)
    -> holds (Some non subject are not object).

Proof.
intros.
eliminateObversion.
eliminateConversion1.
Qed.
```

Note that the tactic `intros` transforms the original form p_1 `->` p_2 such that p_2 remains to be proven (below the bar), where p_1 can be assumed as a premise (above the bar).

In order to express that a rule holds both ways, we use the clause "if and only if", which we often abbreviate as "iff". The following lemma is an example.

Lemma 7 (AllNonNon)**.** *For any terms t_1 and t_2, the proposition* `All non` *t_1* `are non` *t_2 holds iff the proposition* `All` *t_2* `are` *t_1 holds.*

This lemma states the following two rules:

$$\frac{\texttt{All non } t_1 \texttt{ are non } t_2}{\texttt{All } t_2 \texttt{ are } t_1}$$

$$\frac{\texttt{All } t_2 \texttt{ are } t_1}{\texttt{All non } t_1 \texttt{ are non } t_2}$$

Thus, in the proof of the lemma, we need to carry out both directions.

Exercise 7. *Prove both directions, using graphical and text-based notation.*

In Coq, we use the symbol <-> for "iff".

```
Lemma AllNonNon : forall subject object,
   holds (All non subject are non object) <-> holds (All object are subject).

Proof.
intro.
split.
intro.
eliminateObversion.
eliminateConversion2.
eliminateObversion.
intro.
eliminateObversion.
eliminateConversion2.
eliminateObversion.
rewrite NonNon.
rewrite NonNon.
apply H.
Qed.
```

Note that the tactic `split` splits <-> into two directions of ->.

9 Arguments and Syllogisms

An argument consists of a number of propositions, called premises, and a further proposition, called conclusion. For example, Lemma 6 is an argument with one premise, namely `Some non` t_1 `are non` t_2, whose conclusion is `Some non` t_2 `are not` t_1. An argument is valid if in any model in which all premises hold, the conclusion also holds.

A syllogism is an argument with two premises, in which three different terms occur, and in which every term occurs twice, but never twice in the same proposition.

Exercise 8. *Verify that Example 2 contains a syllogism. Show that this syllogism is not valid, using a model as counter-example.*

The following syllogism consists of only universal affirmative propositions.

Axiom 12 (Barbara)**.** *For all terms minor, middle, and major, if* `All` *middle* `are` *major holds, and* `All` *minor* `are` *middle holds, then* `All` *minor* `are` *major also holds.*

The name Barbara for this syllogism is a mnemotic device used for remembering syllogisms. Traditionally, universal affirmative propositions are denoted by the letter 'a', universal negative propositions by the letter 'e', particular affirmative propositions by the letter 'i', and particular negative propositions by the letter 'o'. The vowels in Barbara indicate that all three propositions are universal affirmative.

The Barbara syllogism is graphically depicted as follows:

$$\frac{\texttt{All}\ \textit{middle}\ \texttt{are}\ \textit{major} \qquad \texttt{All}\ \textit{minor}\ \texttt{are}\ \textit{middle}}{\texttt{All}\ \textit{minor}\ \texttt{are}\ \textit{major}}\ [\text{Barbara}]$$

We convince ourselves that Barbara is valid, using the Venn diagram below.



If the first premise holds, then areas 1 and 4 are empty, and if the second premise holds, then areas 2 and 3 are empty. The conclusion simply states that areas 1 and 2 are empty, which clearly follows from the premises, regardless what other properties the model under consideration has.

We can directly apply Barbara for proving `All Greeks are mortal` from Axioms 1 and 2.

Lemma 8. *The proposition `All Greeks are mortal` holds.*

Proof.

$$\frac{\texttt{All Greeks are humans} \qquad \texttt{All humans are mortal}}{\texttt{All Greeks are mortal}}\,[\text{Barbara}]$$

□

In text-based form, the proof reads as follows.

Proof.

1	`All Greeks are humans`	GreeksHumanity
2	`All humans are mortal`	HumansMortality
3	`All Greeks are mortal`	Barbara 1,2

□

Since Barbara has two premises, the justification in Line 3 refers to two other lines, namely Line 1 and Line 2.

In Coq, we are formulating Barbara as an `Axiom` as follows.

```
Axiom Barbara : forall major minor middle,
    holds (All middle are major)
 /\ holds (All minor are middle)
 -> holds (All minor are major).

Lemma GreeksMortality : holds (All Greeks are mortal).

Proof.
apply Barbara with (middle := humans).
split.
apply HumansMortality.
apply GreeksHumanity.
Qed.
```

Note that in order to apply `Barbara`, we need to tell Coq what term to use as `middle`.

10 Fun With Barbara

The English author and logician Lewis Carroll formulated many logical puzzles that can be solved using term logic. Our first example uses the following three premises.

- No ducks waltz.
- No officers ever decline to waltz.
- All my poultry are ducks.

Our version of the puzzle consists of proving from these premises that no officers are my poultry.

Lemma 9 (No-Officers-Are-My-Poutry)**.** *If `No ducks are things-that-waltz` holds, `No officers are non things-that-waltz` holds, and `All my-poutry are ducks` holds, then `No officers are my-poultry` also holds.*

Proof.

1	`No officers are non things-that-waltz`	premise
2	`obvert(All officers are things-that-waltz)`	ObvDef 1
3	`All officers are things-that-waltz)`	ObvE 2
4	`No ducks are things-that-waltz)`	premise
5	`convert(No things-that-waltz are ducks)`	ConvDef 4
6	`No things-that-waltz are ducks`	ConvE_2 5
7	`No things-that-waltz are non non ducks`	NNI 6
8	`obvert(All things-that-waltz are non ducks)`	ObvDef 7
9	`All things-that-waltz are non ducks`	ObvE 8
10	`All my-poultry are ducks`	premise
11	`All my-poultry are non non ducks`	NNI 10
12	`All non non my-poultry are non non ducks`	NNI 11
13	`contrapose(All non ducks are non my-poultry)`	ContrDef 12
14	`All non ducks are non my-poultry`	ContrE_1 13
15	`All things-that-waltz are non my-poultry`	Barbara 9,14
16	`All officers are non my-poultry`	Barbara 3,15
17	`obvert(No officers are my-poultry)`	ObvDef 16
18	`No officers are my-poultry`	ObvE 17

□

We first introduce suitable terms.

```
Parameter ducks : Term.
Parameter things_that_waltz : Term.
Parameter officers : Term.
Parameter my_poultry : Term.
```

The lemma and the proof follow.

```
Lemma No_Officers_Are_My_Poulty :
   holds (No ducks are things_that_waltz) /\
   holds (No officers are non things_that_waltz) /\
   holds (All my_poultry are ducks)
   ->
   holds (No officers are my_poultry).
```

```
Proof.
intro.
destruct H.
destruct H0.
eliminateObversion.
apply Barbara with (middle := things_that_waltz).
split.
apply Barbara with (middle := non ducks).
split.
eliminateContraposition1.
rewrite NonNon.
rewrite NonNon.
apply H1.
eliminateObversion.
rewrite NonNon.
eliminateConversion2.
eliminateObversion.
Qed.
```

Exercise 9. *From the following premises*

- *All babies are illogical.*
- *Nobody is despised who can manage a crocodile.*
- *Illogical persons are dispised.*

prove that no babies can manage a crocodile.

Coq Exercise 4.

```
Parameter babies : Term.
Parameter logical_persons : Term.
Parameter despised_persons : Term.
Parameter persons_who_can_manage_a_crocodile : Term.

Lemma No_baby_can_manage_a_crocodile :
holds (All babies are non logical_persons) /\
holds (No persons_who_can_manage_a_crocodile are despised_persons) /\
holds (All non logical_persons are despised_persons)
->
holds (No babies are persons_who_can_manage_a_crocodile).
```

Complete the proof below.

```
Proof.
admit.
Qed.
```

Exercise 10. *From the following premises*

- *No boys under 12 are admitted to this school as boarders.*
- *All the industrious boys have red hair.*
- *None of the dayboys learn Greek.*
- *None but those under 12 are idle.*

prove that all boys who learn Greek are red-haired.

Coq Exercise 5. *Complete the proof in the following.*

```
Parameter boys_under_12 : Term.
Parameter boys_admitted_as_boarders : Term.
Parameter industrious_boys : Term.
Parameter red_haired_boys : Term.
Parameter boys_who_learn_Greek : Term.

Lemma All_Boys_Who_learn_Greek_are_red_haired :
   holds (No boys_under_12 are boys_admitted_as_boarders) /\
   holds (All industrious_boys are red_haired_boys) /\
   holds (No non boys_admitted_as_boarders are boys_who_learn_Greek) /\
   holds (No non boys_under_12 are non industrious_boys)
   ->
   holds (All boys_who_learn_Greek are red_haired_boys).


Proof.
admit.
Qed.
```

Exercise 11. *From the following premises*

- *No interesting poems are unpopular among people of real taste.*
- *No modern poetry is free from affectation.*
- *All your poems are on the subject of soap-bubbles.*
- *No affected poetry is popular among people of real taste.*
- *No ancient poem is on the subject of soap-bubbles.*

prove that All your poems are non-interesting poems.

Coq Exercise 6. *Complete the proof in the following.*

```
Parameter interesting_poems : Term.
Parameter poems_that_are_popular_among_people_of_real_taste : Term.
Parameter modern_poems : Term.
Parameter affected_poems : Term.
Parameter your_poems : Term.
Parameter poems_on_the_subject_of_soap_bubbles : Term.

Lemma Your_Poems_Are_Not_Interesting :
   holds (No interesting_poems are non poems_that_are_popular_among_people_of_real_taste) /\
   holds (No modern_poems are non affected_poems) /\
   holds (All your_poems are poems_on_the_subject_of_soap_bubbles) /\
   holds (No affected_poems are poems_that_are_popular_among_people_of_real_taste) /\
   holds (No non modern_poems are poems_on_the_subject_of_soap_bubbles)
   ->
   holds (All your_poems are non interesting_poems).

Proof.
admit.
Qed.
```

Exercise 12. *From the following premises*

- *The only animals in this house are cats.*
- *Every animal is suitable for a pet, that loves to gaze at the moon.*
- *When I detest an animal, I avoid it.*
- *No animals are carnivourous, unless they prowl at night.*
- *No cat fails to kill mice.*
- *No animals ever take to me, except what are in this house.*
- *Kangaroos are not suitable for pets.*
- *None but carnivora kill mice.*
- *I detest animals that do not take to me.*
- *Animals, that prowl at night, always love to gaze at the moon.*

prove that I avoid kangaroos.

Coq Exercise 7. *Complete the proof in the following.*

```
Parameter animals_in_this_house : Term.
Parameter things_suitable_for_a_pet : Term.
Parameter animals_that_love_to_gaze_at_the_moon : Term.
Parameter things_that_I_avoid : Term.
Parameter animals_that_I_detest : Term.
Parameter carnivorous_animals : Term.
Parameter things_that_prowl_at_night : Term.
Parameter things_that_kill_mice : Term.
Parameter animals_that_take_to_me : Term.
Parameter kangaroos : Term.



Lemma I_avoid_kangoroos :
holds (All animals_in_this_house are cats) /\
holds (All animals_that_love_to_gaze_at_the_moon are things_suitable_for_a_pet) /\
holds (All animals_that_I_detest are things_that_I_avoid) /\
holds (No carnivorous_animals are non things_that_prowl_at_night) /\
holds (No cats are non things_that_kill_mice) /\
holds (No animals_that_take_to_me are non animals_in_this_house) /\
holds (All kangaroos are non things_suitable_for_a_pet) /\
holds (No things_that_kill_mice are (non carnivorous_animals)) /\
holds (All non animals_that_take_to_me are animals_that_I_detest) /\
holds (All things_that_prowl_at_night are animals_that_love_to_gaze_at_the_moon)
->
holds (All kangaroos are things_that_I_avoid).



Proof.
admit.
Qed.
```

11 Other Syllogisms

The syllogism under consideration in the previous sections only handles universal affirmative propositions. The following syllogism

Axiom 13 (Celarent[3])**.** *For all terms minor, middle, and major, if* `No` *middle* `are` *major holds, and* `All` *minor* `are` *middle holds, then* `No` *minor* `are` *major also holds.*

[3]The vowels 'e', 'a' and 'e' in the word "Celarent" indicate that the first premise and the conclusion are universal negative ('e') and the second premise is universal positive ('a').

Exercise 13. *Use a Venn diagram to convince yourself of the validity of Celarent.*

Axiom 14 (Darii)**.** *For all terms minor, middle, and major, if* `All` *middle* `are` *major holds, and* `Some` *minor* `are` *middle holds, then* `Some` *minor* `are` *major also holds.*

Exercise 14. *Use a Venn diagram to convince yourself of the validity of Darii.*

The three syllogisms Barbara, Celarent and Darii have the following structure in common: The first premise shares its object with the conclusion, and the second premise shares its subject with the conclusion, and no complement is used in any proposition. Such syllogisms are called *Figure 1* syllogisms.

Exercise 15. *How many Figure 1 syllogisms (valid or not) exist?*

Exercise 16. *In addition to Barbara, Celarent and Darii, there is one other valid Figure 1 syllogism. Can you find it? Give it a name, according to the mnemonic device used for the other three.*